Electroacoustic Simulation for Neophytes

Mark Kahrs
(mark.kahrs@gmail.com)

1. Introduction

2. Components

    - Electrical
    - Mechanical
    - Acoustical
    - Interconnections

3. Simulation with SPICE

4. Applications

    - Loudspeakers
    - Horns
    - Headphones (next time)

5. Conclusion

Introduction

1. Lumped vs. Distributed Parameters

2. Linear vs. Non-linear models

3. Reasoning by Analogy

   - EFP (Voltage-Force-Pressure)

   - IFP (Current-Pressure-Force)

Resistance

- Electrical:

$$r_E = \frac{e}{i} \tag{1}$$

- Mechanical:

$$r_M = \frac{f_M}{v} \tag{2}$$

- Acoustical:

$$r_A = \frac{p}{U} \tag{3}$$

Capacitance/Compliance

- Electrical:

$$i = C_E \frac{de}{dt} \tag{4}$$

$$C_E = \frac{q}{e} \tag{5}$$

- Mechanical:

$$C_M = \frac{x}{f_M} \tag{6}$$

- Acoustical:

$$C_A = \frac{V}{\rho c^2} \tag{7}$$

Inertance / Inductance

- Electrical:

$$e = L_E \frac{di}{dt} \tag{8}$$

- Mechanical (mass):

$$f_M = m \frac{dv}{dt} \tag{9}$$

- Acoustical (inertance):

$$p = M \frac{dU}{dt} \tag{10}$$

$$M = \frac{m}{S^2} \tag{11}$$

Summary of Domains and units:

Electrical:

Resistance	ohms
Capacitance	farads
Inductance	henries

Mechanical:

Resistance	Mech. ohms
Compliance	m/n
Mass	kg

Acoustical:

Resistance	rayls
Compliance	m^5-n
Mass	kg-m^4

Summary of Components:

Electrical	Mechanical	Acoustical
Resistor	Friction	Ac. Resistance
Capacitor	Spring	Volume
Inductor	Mass	Tube

Domain interconnection: Electromagnetic-mechanical

- Back emf:

$$e = Blu \tag{12}$$

- Force generated:

$$f = Bli \tag{13}$$

Solution: Controlled Sources (CCVS, CCCS)

Domain interconnection: Mechanical-acoustical

- Force over area:

$$f = pS \tag{14}$$

- mechanical velocity:

$$v = \frac{U}{S} \tag{15}$$

$$Z_M = \frac{f}{v} = \frac{pS}{\frac{U}{S}} = \frac{p}{U}S^2 \tag{16}$$

Solution: Controlled Sources (VCVS, CCCS)

Simulation with circuit simulators

- Use circuit simulators (like SPICE)
- Public domain software (i.e., free)
- *linear circuits* only
- Can perform transient, AC analyses

Strategy

- Describe each domain with R,L,Cs
- Interconnect with controlled sources
- Insert "meters" at interesting points
- Label with node numbers
- Perform transient, AC analyses

What is SPICE?

- Simulation Program with Integrated Circuits Emphasis
- Open Source code from Berkeley
- Available with various enhancements: PSPICE, LTSPICE
- Syntax dates back to the punched card era (really)

SPICE description

- Title Card Description of circuit
- Comment Card a line beginning with the symbol *
- Element Cards circuit is described with element cards: element name and the node numbers
- Control Cards Specify options to the program and the analyses to be performed. All control cards begin with the character '.'
- the .END card

Analysis methods

- Frequency response (incl. Bode Plots)

- Transient Response

- DC analysis (not useful)

- Noise analysis (not useful)

SPICE syntax - elements

Resistance	Rx node node value
Capacitor/Compliance	Cx node node value
Inductance/Inertance	Lx node node value
Voltage/Pressure	Vx node+ node- value
Current/Velocity	Ix node+ node- value

SPICE syntax - controlled sources

VCVS	Ex node+ node- control+ control- gain
VCCS	Gx node+ node- control+ control- gain
CCCS	Fx node+ node- y gain
CCVS	Hx node+ node- y gain

Electroacoustic variables

e	V
i	I
f	V
v	I
p	V
U	I

Analysis strategy:

- All volumes become acoustic compliances (C) with one terminal grounded

- Connect compliances with pipes (L) and resistors (R)

- Leaks are resistors (R) to ground

- Outside enclosure is *radiated* to radiation impedance

Acoustic components:

- Cavities: $Z_A = \frac{V}{\rho c^2}$

- Pipes: $Z_A = \frac{V}{\rho c^2}$

- Resistances: depends...

- Radiation impedance: (approx. from Beranek)

$$Z_{AL} = \frac{\rho_0 \omega^2 \pi a^4}{2c_0} + j\frac{8\rho_0 c_0^2 a}{3} \tag{17}$$



Electrodynamic loudspeaker: US Patent 1,812,389



	cone	12
	former	36
	voice coil	22
Parts:	magnetic annulus	23
	spider	40
	chassis	36
	centering ring	39

Modeling the loudspeaker



Force factor $= Bli$

Back emf $= Blv$

Diaphragm area $= S_D$

Boxes with ports

- Box volume = compliance
- Box edges (leaks) = resistance
- Back of diaphragm air = mass
- Port = mass
- Port couples with the driver

Ported box



Ported box parameters

Resistance	R_E	6.5Ω
Inductance	L_E	1 mH
Damping	R_{MD}	3.72
Suspension compliance	C_{MD}	$2.14e-4$
Diaphragm mass	M_{MD}	0.064
12" Driver area	S_D	4.5e-2
38mm port	S_P	
Box compliance	C_{AB} : $V_{AB}/\rho c^2$	4.03e-7
Air load on diaphragm	M_{AB}	2.03
Box leaks	R_{AB}	6.91e-4
Driver port coupling	k_{DP}: $3\pi a/16d$	0.393
Port driver coupling	k_{PD}	0.124

SPICE input - Electro-mechanical

```
Vented box (from Leach)
Veg 1 0 AC 1V DC 0V
Rew 1 2 6.5
Lew 2 3 1M
HBLUW 3 4 V2W 16.5
V1W 4 0 AC 0V DC 0V
*
HBLIW 5 0 V1W 16.5
LMmdw 5 6 0.064
Rmsw 6 7 3.72
Cmsw 7 8 2.14e-4
ESDPW 8 9 17 10 4.5E-2
V2W 9 0 AC 0V DC 0V
```

SPICE input - Acoustical

```
FSDUW 10 17 V2W 4.5e-2
Lmabw 10 11 2.03
Cabw 11 12 4.03e-7
V3W 12 0 AC 0V DC 0V
Ralw 11 0 6.91E4
Lmap 11 13 30.9
*
FKPUW 15 13 V6W 0.124
Lma1p 13 15 8.37
Ra1p 13 14 3.94e4
ca1p 13 14 2.34e-9
Ra2p 14 16 8.93e4
V4W 15 16 AC 0V DC 0V
V5W 16 0 AC 0V DC 0V
*
FKPUP 19 17 V4W 0.393
Lma1w 17 19 2.66
Ra1w 17 18 3.97e3
ca1w 17 18 7.31e-8
Ra2w 18 20 9e3
V6W 19 20 AC 0V DC 0V
V7W 20 0 AC 0V DC 0V
```

SPICE input - control

```
.AC DEC 20 10 10K
.print ac I(V7W)
.END
```

How to use the results...

$$|p(r)| = \frac{\rho_0 f}{r} |U_B| \tag{18}$$

$$\mathsf{SPL} = 20 \log_{10} p/p_{\mathsf{ref}} \tag{19}$$

Where: $p_{\mathsf{ref}} = 10^{-5}$ Pa, $\rho_0 = 1.18$
$|U|$? Insert Ammeters at important branches...
Let $r = 1$.

Simulation



Loudspeaker simulations

Compression driver



(from U.S. Patent 1,707,545)

Compression driver parts		
	magnet	11
	voice coil	14
	diaphragm	17
	phasing plug	23
	back cavity	27
	resistor	28

Phasing plug



(from U.S. Patent 5,117,462)

Modeling the compression driver



What is Z_{AB}? What is Z_{AF}? What are the electrical and mechanical parameters?

Compression driver model

- Z_{AB} = back cavity compliance

- Z_{AF} = front cavity, phasing plug, load

Resistance	R_E	7.5Ω
Inductance	L_E	0.35 mH
Force factor	Bl	19 N-m
Resonance	f_s	560 Hz
Damping	R_{MD}	unknown
Suspension compliance	C_{MD}	unknown
Diaphragm mass	M_{MD}	0.0032 gm
Driver area	S_D	0.002

Mechanical damping and compliance are not given.

Nonlinear model



Additional calculations:

- $F_S = \frac{1}{2\Pi\sqrt{C_{MD}M_{MD}}}$ so $C_{MD} = 2.52e-5$

- R_{MD} is still unknown

- $C_{AB} = 7.04e-6V_{AB}$, what is V_{AB}???
  Let $V_{AB} = 1.2e-4$, then $C_{AB} = 8.4480e-10$

- $C_{AF} = 7.04e-6V_{AF}$, what is V_{AF}???
  Let $V_{AF} = 2.9368e{-}04$, then $C_{AF} = 2.0675e{-}09$

- $M_{MP} = \rho V_P/3 = 1.2131e-05$

SPICE input - Electro-mechanical

```
JBL 2446H
Veg 1 0 AC 1V DC 0V
Re 1 2 7.5
Le 2 3 0.35m
HBLU 3 4 V2 19
V1 4 0 AC 0V DC 0V
*
HBLI 5 0 V1 19
LMms 5 6 3.2m
Rms 6 7 1e-6
Cms 7 8 2.52e-5
ESDP 8 9 20 10 0.008
V2 9 0 AC 0V DC 0V
```

SPICE input - Acoustical

```
FSDU 10 20 V2 0.008
* back cavity
CAB 10 11 8.4480e-10
Rab 10 11 1e12
V3 11 0 AC 0V DC 0V
* front cavity
CAF 20 30 2.0675e-09
* phase plug
LAP 20 23 2.0675e-06
RAP 23 24 126
V5 24 21 AC 0V DC 0V
* 2 inch air load
la1 21 30 6.2515
ca1 21 22 5.6758e-09
ra1 21 22 2.1844e+04
ra2 22 30 4.9532e+04
V4 30 0 AC 0V DC 0V
```

Simulation



Horn simulation

SPICE simulation deficiencies

- Lumped circuits only (not distributed)
- Linear circuits only (or hack solutions)
- Can't study wave phenomena

Where to read more...

- Beranek, *Acoustics*
  An oldie but goodie. Available from ASA (in pb).

- Olson, *Dynamic Analogies*
  Another oldie but real goodie (on the web)

- Leach, *Introduction to Electroacoustics and Audio Amplifier Design*
  a Modern master

Next talk... How to use SPICE to simulate various headphones and earphones

July 24th?